///////////////////////////////////////////////////////////////////////////////
// East Baton Rouge
///////////////////////////////////////////////////////////////////////////////

*******************************************************************************
* Specify city and years
*******************************************************************************

local city = "east_baton_rouge"

use dataset_`city'_cross, clear

local y1 2000
local y2 2010
local y3 2020

*******************************************************************************
* Histograms
*******************************************************************************

foreach y in `y1' `y2' `y3' {

	su pctyes if year == `y' [aw=totalvotes]
	local mymn = string(r(mean), "%20.02fc")
	local mysd = string(r(sd), "%20.02fc")
	local mysubtitle = "Mean = `mymn', S.D. = `mysd'"
	
	
	#delimit;

	gr tw
		(hist pctyes if year == `y', freq width(0.05) start(0)
			fcol(gray) lcol(black) lwid(thin) barwidth(0.05))
		,
			plotregion(style(none))
			ylab(, angle(horiz))
			name(g`y', replace)
			xline(0.5, lpat(dash) lwid(thick))
			xlab(0(0.25)1)
			title(`y', size(huge))
			subtitle("`mysubtitle'")
			ytitle("")
			xtitle("")
			scale(1.5)
			;
	
	#delimit cr
	
	gr close
}

gr combine g`y1' g`y2' g`y3', xcommon rows(1) ysize(1) xsize(3) ///
	b1title("Share Yes Vote") l1title("Number of Precincts")
	
gr export "_output/figure 1 a.eps", replace

///////////////////////////////////////////////////////////////////////////////
// Fort Worth
///////////////////////////////////////////////////////////////////////////////

*******************************************************************************
* Specify city and years
*******************************************************************************

local city = "fort_worth"

use dataset_`city'_cross, clear

local y1 2009
local y2 2014
local y3 2020

*******************************************************************************
* Histograms
*******************************************************************************

foreach y in `y1' `y2' `y3' {

	su pctyes if year == `y' [aw=totalvotes]
	local mymn = string(r(mean), "%20.02fc")
	local mysd = string(r(sd), "%20.02fc")
	local mysubtitle = "Mean = `mymn', S.D. = `mysd'"
	
	#delimit;

	gr tw
		(hist pctyes if year == `y', freq width(0.05) start(0)
			fcol(gray) lcol(black) lwid(thin) barwidth(0.05))
		,
			plotregion(style(none))
			ylab(, angle(horiz))
			name(g`y', replace)
			xline(0.5, lpat(dash) lwid(thick))
			xlab(0(0.25)1)
			title(`y', size(huge))
			subtitle("`mysubtitle'")
			ytitle("")
			xtitle("")
			scale(1.5)
			;
	
	#delimit cr
	
	gr close
}

gr combine g`y1' g`y2' g`y3', xcommon rows(1) ysize(1) xsize(3) ///
	b1title("Share Yes Vote") l1title("Number of Precincts")
	
gr export "_output/figure 1 b.eps", replace

///////////////////////////////////////////////////////////////////////////////
// End
///////////////////////////////////////////////////////////////////////////////
